Intention, beliefs and mood assessed using electronic diaries predicts attendance at cardiac rehabilitation

Jones, Martyn; Smith, Karen; Herber, Oliver; White, Myra; Steele, Fiona; Johnston, Derek W.

Published in:
International Journal of Nursing Studies

DOI:
10.1016/j.ijnurstu.2018.08.015

Publication date:
2018

Document Version
Peer reviewed version

Supplementary Materials: Further Details of Statistical Methods

1. Multilevel structural equation models for effects of baseline and changes in cardiac-related beliefs and mood on attendance

The results given in **Tables 1-4 (main paper)** are from a series of multilevel SEMs consisting of a growth curve model for change in a cardiac-related belief or mood, and a model for subsequent cardiac rehabilitation (CR) attendance with baseline and change in belief or mood as predictors. Figure 1 shows the corresponding path diagram for the model. For each belief/mood variable a random slopes linear growth model was estimated simultaneously with a logistic regression model for attendance, where the attendance model included as predictors the latent intercept and slope variables from the growth model. For each belief or mood variable, estimates from the growth part of the model are given in Table 1a and estimates from the logistic model for attendance are given in Table 1b. The model for each belief/mood was then extended to include a set of demographic controls in the attendance model (**Table 2 main paper**). All analysis was carried out using the free aML program (Lillard and Panis, 1998-2003).



Figure 1. Path diagram for multilevel SEM with linear random slopes growth model for change in belief or mood (B) and effects of baseline belief/mood (I) and rate of change in belief/mood (S) on subsequent CR attendance

2. Multilevel structural equations models with mediating effects of "do not intend" in relationship between attendance and other cardiac-related beliefs and mood

The SEM of Figure 1 was modified to test for mediating effects of "do not attend" (DNI) in the relationship between selected cardiac-belief and mood variables (X) and CR attendance. Two time-varying 'X' variables were considered in turn: perceived necessity and negative affect.

For each X variable the following multilevel SEMs were fitted. Selected results from these models are presented in **Table 3** for perceived necessity and **Table 4** for negative affect.

- **Model 1.** SEM allowing for a direct effect of X on attendance, excluding the effect of DNI on attendance.
- **Model 2.** SEM with direct effects of X and DNI on attendance. The model includes effects of both baseline DNI (the intercept random effect from the growth curve model) and change in DNI (slope random effect). The growth models for DNI and X and the model for attendance are estimated simultaneously.
- **Model 3.** Mediation model with direct effects of X and DNI (intercept and slope) on attendance, and indirect effect of X on attendance via DNI.

The full SEM (Model 3) is shown in Figure 2. As the models are nested, they can be compared using likelihood ratio tests. The analysis was carried out using Mplus (Muthén and Muthén, 1998-2010). Although it is possible to estimate mediation models in aML it is more straightforward to specify, and quicker to estimate, general SEMs such as Model 3 in Mplus. Annotated Mplus syntax for Model 3 is provided below.

Model 3 extends Model 2 in two ways: (i) direct effects of the observed variable X at week t on DNI at t are estimated, and (ii) correlations are estimated among the random effects for X and DNI. The path diagram for Model 3 with a random intercept model fitted for X is shown in Figure 2. These extensions allow for mediation effects of DNI at the week and individual levels. In the negative affect analysis, for example, with random slope models estimated for both X and DNI, (ii) involves estimation of the four correlations among the intercepts and slopes. To illustrate their interpretation, consider the positive correlation between the NA intercept and the DNI slope (see **Table 4**); this implies that a higher-than-average negative affect at baseline is associated with a faster-than-average decline in DNI.



Figure 2. Path diagram for multilevel SEM with mediation effects of "do not intend" (DNI) in the relationship between a time-varying cardiac-belief or mood variable (X) and attendance at cardiac rehab (Model 3). The model for time-varying X is a random intercept growth model with individual-specific intercept I_X. The model for DNI is a random slope growth model with individual-specific intercept I_{DNI} and slope S_{DNI}.

Mplus syntax for mediation model (Model 3)

The syntax below is for the mediation model of Figure 2 where the X variable is perceived necessity. The input data are in 'wide' form with one record per patient and 15 weekly values of 'do not intend' (DNI) and standardised perceived necessity (ZPNEC) stored as separate variables. The binary indicator of attendance at CR (ATTEND) is declared as a categorical variable. By default a logistic model is fitted for ATTEND when specified as a dependent variable.

```
Data:
  File is dni_attend.dat;
Variable:
  Names are
     dni1 dni2 dni3 dni4 dni5 dni6 dni7 dni8 dni9 dni10 dni11
     dni12 dni13 dni14 dni15 zpnec1 zpnec2 zpnec3 zpnec4 zpnec5 zpnec6
     zpnec7 zpnec8 zpnec9 zpnec10 zpnec11 zpnec12 zpnec13 zpnec14 zpnec15
     attend;
  Categorical = attend;
  Missing are all (-9999) ;
  Usevariables = dni1 dni2 dni3 dni4 dni5 dni6 dni7 dni8 dni9 dni10 dni11
     dni12 dni13 dni14 dni15 zpnec1 zpnec2 zpnec3 zpnec4 zpnec5 zpnec6
     zpnec7 zpnec8 zpnec9 zpnec10 zpnec11 zpnec12 zpnec13 zpnec14 zpnec15
```

```
    attend;
```

Random effects models are specified using the 'random' analysis type. A linear random slope model is fitted for DNI with equality constraints on the intercepts over time. The random intercept and slope are named IDNI and SDNI respectively. The random intercept and slope variances, and their covariance, are declared explicitly and given labels (for computing functions of these and other parameters later). A random intercept model is fitted for ZPNEC; although a slope random effect 'spnec' is specified, its variance is constrained to zero.

```
Analysis:
  Type = random;
Model:
  idni sdni | dni1@0 dni2@1 dni3@2 dni4@3 dni5@4 dni6@5 dni7@6 dni8@7
    dni9@8 dni10@9 dni11@10 dni12@11 dni13@12 dni14@13 dni15@14;
  dni1-dni15 (1);
  idni WITH sdni (cov_isdni);
  idni (vidni);
  sdni (vsdni);
  ipnec spnec | zpnec1@0 zpnec2@1 zpnec3@2 zpnec4@3 zpnec5@4 zpnec6@5
     zpnec7@6 zpnec8@7 zpnec9@8 zpnec10@9 zpnec11@10 zpnec12@11
     zpnec13@12 zpnec14@13 zpnec15@14;
  ipnec (vipnec);
  spnec@0;
  idni WITH ipnec (cov_idipn);
  sdni WITH ipnec (cov_sdipn);
  zpnec1-zpnec15 (2);
```

The next part of the syntax specifies a logistic model for ATTEND with the intercept and slope random effects from the growth model for DNI and the intercept random effect from the growth for ZPNEC as predictors.

```
  attend ON idni (b_idni);
  attend ON sdni (b_sdni);
  attend ON ipnec (b_ipnec);
```

A direct effect of ZPNEC at week t on DNI at t is fitted, assuming a constant effect across time.

```
  dni1 ON zpnec1 (3);
  dni2 ON zpnec2 (3);
  dni3 ON zpnec3 (3);
  dni4 ON zpnec4 (3);
  dni5 ON zpnec5 (3);
  dni6 ON zpnec6 (3);
  dni7 ON zpnec7 (3);
  dni8 ON zpnec8 (3);
  dni9 ON zpnec9 (3);
  dni10 ON zpnec10 (3);
  dni11 ON zpnec11 (3);
  dni12 ON zpnec12 (3);
  dni13 ON zpnec13 (3);
  dni14 ON zpnec14 (3);
  dni15 ON zpnec15 (3);
```

Several new parameters are defined as functions of the model parameters. The first set of new parameters are random effect correlations between the intercept and slope of DNI (COR_ISDNI), the intercepts of DNI and ZPNEC (COR_IDIPN), and the slope of DNI and intercept of ZPNEC (COR_SDIPN). The second set of new parameters are standardised regression coefficients in the logistic model for attendance; the effects of the latent random intercept and slope for DNI and random intercept for PNEC are multiplied by the corresponding standard deviation.

```
Model constraint:
  NEW(cor_isdni cor_idipn cor_sdipn sb_idni sb_sdni sb_ipnec);
  cor_isdni = cov_isdni/(SQRT(vidni)*SQRT(vsdni));
  cor_idipn = cov_idipn/(SQRT(vidni)*SQRT(vipnec));
  cor_sdipn = cov_sdipn/(SQRT(vsdni)*SQRT(vipnec));
  sb_idni = b_idni*SQRT(vidni);
  sb_sdni = b_sdni*SQRT(vsdni);
  sb_ipnec = b_ipnec*SQRT(vipnec);
```

References

Lillard, L. A. and Panis, C. W. A. (1998-2003) *aML User's Guide and Reference Manual, Version 2*, Los Angeles: EconWare. Download from http://applied-ml.com/index.html

Muthén, L.K. and Muthén, B.O. (1998-2010). *Mplus User's Guide*. Sixth Edition. Los Angeles, CA: Muthén & Muthén.